PharmaSUG 2012 - Paper SP08

USING SAS® TO CALCULATE NONCOMPARTMENTAL URINE PARAMETERS

Vanessa Rubano, Boehringer Ingelheim Pharmaceuticals Inc., Ridgefield, CT
Modesta Wiersema, Boehringer Ingelheim Pharma GmbH & Co. KG., Biberach, DE

ABSTRACT

Clinical Pharmacology is an integral part of clinical trials and the approval of a new drug. To obtain regulatory approval for a New Drug Application (NDA), a sponsor is required to show the elimination of drug and metabolites within the human body's matrices, such as urine and feces, supported by plasma pharmacokinetic (PK) endpoints. The industry standard used for calculating PK endpoints is the application of clinical trial data to WinNonLin® (WNL), a software modeling package that is based on mathematical equations. The authors of this paper aim to provide an introduction to PK urine assessments and describe a method, using SAS® programming, to derive PK endpoints for individual and cumulative amounts excreted (Ae), fractions of the administered dose excreted (fe) and renal clearance (CLR) of drug from the plasma matrix to urine for a complete profile of a subject or patient. The methods outlined within this paper were implemented using DATA step programming elements in conjunction with user defined macros.

INTRODUCTION

Clinical Pharmacokinetics and Pharmacodynamics (PK/PD) is the study of the disposition of a drug and the effects of the drug and metabolites on the body, specifically the movement of drugs within various biological systems, to the effective receptors as affected by the absorption, distribution, metabolism, and elimination (ADME) pathways. Pharmacokinetics is an integral part of clinical trials of an investigational new drug, as there are correlations between drug concentrations and their pharmacologic responses. This drug concentration data is captured for PK samples for a particular study as defined within its clinical trial protocol. Programming of the PK analysis data sets is outside the scope of this paper, however the assignment of treatments, visits, analytes, and bioanalytical data are assumed to be provided. The key variables pertaining to this paper and the generation of PK urine endpoints are planned sampling start and stop times, matrices, visits, patient numbers, treatments, urine volumes, urine concentrations of analyte, actual amount dosed, imputed assignments of qualifers for missing data and plasma WNL noncompartmental analysis, such as area under the curve (AUC).

Because of the myriad biological processes at work to alter drug concentrations in tissues and fluids, the handling of a drug by the body is complex. Simplifications of these bodily processes are necessary to model drug behavior. By applying PK mathematical principles to clinical data, one can develop a basic type of model for the body used to describe many of the body's compartmental processes involved in drug movement. The excretion of drug from plasma into urine as a function of sampling time intervals allows for the generation of kidney-based clinical endpoints. The clinical endpoints determining the renal elimination of a drug are the amount excreted (Ae) in the urine, and the fraction of drug excreted (fe) in the urine, and renal clearance (CLR). The calculations to derive these endpoints for a given interval (t1-t2) are provided below.

- Ae t1-t2 = urine concentration * urine volume
- Fet1-t2 = Aet1-t2 / dose * 100
- CLRt1-t2 = Aet1-t2 / AUCt1-t2
- Ae t0-t2$^{\#}$ = urine concentration * urine volume
- Fet0-t2$^{\#}$ = Aet0-t2 / dose * 100
- CLRt0-t2$^{\#}$ = Aet0-t2 / AUCt0-t2

$^{\#}$cumulative

METHOD

The primary method for calculating PK urine values using a SAS® program template is illustrated in the diagram below, Figure 1: Overview of Calculating Urine Parameters using SAS®. A SAS® program was developed based on a standardized PK data file structure in which key variables were identified for the calculation of PK urine endpoints.



Figure 1. Overview of Calculating Urine Parameters Using SAS®

SETTING UP A WORKING DATA FILE

The initial step involves importing the plasma NCA results file generated by WNL and the urine rawdata file. Since PK data files are typically provided as a PRN file, the programmer creates a new data file, *urine*, from the raw data file, *urine_rawdata*. The authors accomplished this by reading in the rawdata as a delimited file, supplied the PRN formats for the rawdata file's variables and assigned SAS® file formats using the INFILE statement and DELIMITER.

```
DATA urine;
  INFILE '\\Urine_rawdat.prn'
  DELIMITER='09'x MISSOVER DSD LRECL=32767 FIRSTOBS=2 ;
  INFORMAT studyid      $15. ;
  INFORMAT usubjid       10. ;
  INFORMAT ...         10. ;   /*Additional variables*/
  FORMAT   studyid      $15. ;
  FORMAT   usubjid       10. ;
  FORMAT   ...         10. ;  /*Additional variables*/
  INPUT    studyid $
           usubjid
           ...               /*Additional variables*/
      ;
RUN;
```

This technique used for the urine rawdata file is also employeed when importing WNL noncompartmental AUC results. The purpose of importing plasma results is contigent on the clinical trial protocol endpoints requesting the renal clearance of an analyte. Both files are then treated to a series of data steps sorted and merged by treatment, visit number, patient number and planned time. Samples taken before drug administration are deleted to create a working data file. The identifiers for a urine sample are the intervals given as start and stop time (t1-t2). These intervals are then translated into PK urine endpoint names by concattenating the type of endpoint, i.e. Ae, to the sampling interval, i.e. 0_4 and create a urine endpoint name, i.e. Ae0_4. Each of the variables created in the file *kpnm*, refer to a type of PK urine endpoint, i.e. Aet1-t2, fet1-t2, CLRt1-t2, Aet0-t2,…..

```
DATA intervals;
  SET urinesrt;
  LENGTH rawint intnm $20;
  /*Creating variable for urine planned start and stop sampling times*/
rawint=INPUT(TRIM(LEFT(PUT(ptmn,10.)))||'_'||TRIM(LEFT(PUT(ptmspn,10.))
),$20.);

/*Translating urine intervals into a PK urine endpoint name*/
  IF rawint EQ '336_340' THEN intnm= '0_4_ss';
  ELSE IF rawint EQ '340_348' THEN intnm= '4_12_ss';
  ELSE IF rawint EQ '348_360' THEN intnm= '12_24_ss';

/*Deleting pre-dose data*/
  ELSE IF rawint EQ '334_336'   THEN DELETE;

RUN;
```

```
DATA kpnms;
  SET intervals;
  FORMAT aet1_t2 fet1_t2 auct1_t2 clrt1_t2 ae0_t2 fe0_t2 auc0_t2
clr0_t2 $20.;
  aet1_t2  = 'Ae'||TRIM(LEFT(intnm));
  fet1_t2  = 'fe'||TRIM(LEFT(intnm));
  auct1_t2 = 'AUC'||TRIM(LEFT(intnm));
  clrt1_t2 = 'CLR'||TRIM(LEFT(intnm));

  IF INDEX(intnm, '0_4')=0 THEN DO;
    ae0_t2  ='Ae0'||TRIM(LEFT(SUBSTR(intnm,(INDEX(intnm, '_')))));
    fe0_t2  ='fe0'||TRIM(LEFT(SUBSTR(intnm,(INDEX(intnm, '_')))));
    auc0_t2 ='AUC0'||TRIM(LEFT(SUBSTR(intnm,(INDEX(intnm, '_')))));
    clr0_t2 ='CLR0'||TRIM(LEFT(SUBSTR(intnm,(INDEX(intnm, '_')))));
  END;
RUN;
```

Kpnm data file result

INTNM	AET1_T2	FET1_T2	AUCT1_T2	CLRT1_T2	AE0_T2	FE0_T2	AUC0_T2	CLR0_T2
0_4_ss	Ae0_4_ss	fe0_4_ss	AUC0_4_ss	CLR0_4_ss				
4_12_ss	Ae4_12_ss	fe4_12_ss	AUC4_12_ss	CLR4_12_ss	Ae0_12_ss	fe0_12_ss	AUC0_12_ss	CLR0_12_
12_24_ss	Ae12_24_ss	fe12_24_ss	AUC12_24_ss	CLR12_24_ss	Ae0_24_ss	fe0_24_ss	AUC0_24_ss	CLR0_24_
0_4_ss	Ae0_4_ss	fe0_4_ss	AUC0_4_ss	CLR0_4_ss				
4_12_ss	Ae4_12_ss	fe4_12_ss	AUC4_12_ss	CLR4_12_ss	Ae0_12_ss	fe0_12_ss	AUC0_12_ss	CLR0_12_
12_24_ss	Ae12_24_ss	fe12_24_ss	AUC12_24_ss	CLR12_24_ss	Ae0_24_ss	fe0_24_ss	AUC0_24_ss	CLR0_24_
0_4_ss	Ae0_4_ss	fe0_4_ss	AUC0_4_ss	CLR0_4_ss				
4_12_ss	Ae4_12_ss	fe4_12_ss	AUC4_12_ss	CLR4_12_ss	Ae0_12_ss	fe0_12_ss	AUC0_12_ss	CLR0_12_
12_24_ss	Ae12_24_ss	fe12_24_ss	AUC12_24_ss	CLR12_24_ss	Ae0_24_ss	fe0_24_ss	AUC0_24_ss	CLR0_24_
0_4_ss	Ae0_4_ss	fe0_4_ss	AUC0_4_ss	CLR0_4_ss				
4_12_ss	Ae4_12_ss	fe4_12_ss	AUC4_12_ss	CLR4_12_ss	Ae0_12_ss	fe0_12_ss	AUC0_12_ss	CLR0_12_
12_24_ss	Ae12_24_ss	fe12_24_ss	AUC12_24_ss	CLR12_24_ss	Ae0_24_ss	fe0_24_ss	AUC0_24_ss	CLR0_24_
0_4_ss	Ae0_4_ss	fe0_4_ss	AUC0_4_ss	CLR0_4_ss				
4_12_ss	Ae4_12_ss	fe4_12_ss	AUC4_12_ss	CLR4_12_ss	Ae0_12_ss	fe0_12_ss	AUC0_12_ss	CLR0_12_
12_24_ss	Ae12_24_ss	fe12_24_ss	AUC12_24_ss	CLR12_24_ss	Ae0_24_ss	fe0_24_ss	AUC0_24_ss	CLR0_24_
0_4_ss	Ae0_4_ss	fe0_4_ss	AUC0_4_ss	CLR0_4_ss				
4_12_ss	Ae4_12_ss	fe4_12_ss	AUC4_12_ss	CLR4_12_ss	Ae0_12_ss	fe0_12_ss	AUC0_12_ss	CLR0_12_
12_24_ss	Ae12_24_ss	fe12_24_ss	AUC12_24_ss	CLR12_24_ss	Ae0_24_ss	fe0_24_ss	AUC0_24_ss	CLR0_24_

In parallel, a series of data steps filter through the plasma NCA results to select results (i.e. AUC0_4) required for the derivation of clearance values (if needed). The urine data is merged with AUC values in files *aucadd_pre* and *aucadd*.

CALCULATING PK URINE ENDPOINTS

The *values* data file now contains all the required variables from the urine rawdata and WNL results file. The PK urine clinical endpoints are derived using the equations, given in the introduction, and basic operators in SAS®. Additionally, it is important to be mindful of units for the endpoint when writing the SAS® code. This means a few extra programming steps may be needed if the final values for the endpoint need to be converted in the equations.

```
DATA values;
  SET addauc;
  aet1_t2_v=ura; /*Ae values*/
  aet1_t2_u='ng'; /*Ae uints*/

  fet1_t2_v=(aet1_t2_v/(admda*1000*1000))*100;
  /*fe values, included calculation for conversion*/
  fet1_t2_u='%'; /*fe units*/

  clrt1_t2_v=(aet1_t2_v/auct1_t2_v)/60;
 /*Renal CLR values, included calculation for conversion*/
  clrt1_t2_u='mL/min';  /*Renal CLR units*/
RUN;
```

Values data file result

AUCT1_T2_V	AUCT1_T2_U	AET1_T2_V	AET1_T2_U	FET1_T2_V	FET1_T2_U	CLRT1_T2_V	CLRT1_T2_U
471.81171195	hr*ng/mL	23241.2	ng	0.0464824	%	0.8209913479	mL/min
870.88285591	hr*ng/mL	37810	ng	0.07562	%	0.7235952142	mL/min
1180.1091517	hr*ng/mL	47400	ng	0.0948	%	0.6694296022	mL/min
544.88745381	hr*ng/mL	23806.7	ng	0.0476134	%	0.7281840141	mL/min
1078.0809495	hr*ng/mL	39560	ng	0.07912	%	0.6115805438	mL/min
1510.3905471	hr*ng/mL	37261.8	ng	0.0745236	%	0.4111718	mL/min
77.053970472	hr*ng/mL	1784.76	ng	0.00356952	%	0.3860411062	mL/min
155.30092894	hr*ng/mL	4700	ng	0.0094	%	0.5043970688	mL/min
229.44967029	hr*ng/mL	8791.2	ng	0.0175824	%	0.6385714123	mL/min
802.84779932	hr*ng/mL	42320	ng	0.04232	%	0.8785392872	mL/min
1493.9630888	hr*ng/mL	65209.6	ng	0.0652096	%	0.7274789282	mL/min
2123.6516004	hr*ng/mL	105020	ng	0.10502	%	0.8242092691	mL/min
3195.4507794	hr*ng/mL	134292	ng	0.134292	%	0.7004332579	mL/min
6301.7288773	hr*ng/mL	241240	ng	0.24124	%	0.6380259679	mL/min
9864	hr*ng/mL	204782	ng	0.204782	%	0.3460090565	mL/min
941.69146404	hr*ng/mL	103622	ng	0.103622	%	1.8339694043	mL/min

Once individual PK urine endpoints are available a user defined macro, *cumulativ*, is used to calculate cummulative PK endpoints for Ae and fe values. Just as in the individual endpoint calculations, only observations for summation are retained. The cumulative values are the summutation of individual Ae and fe values for patient profile. Missing data and observations that are excluded have rules applied to the derivations. For example, when a missing value is present for an interval, the subsequent calculations for cumulative intervals are not calculated. Units of cummulative endpoints are taken from individual units.

```
%MACRO cumulativ(indata=, inparam=);
DATA cum&inparam.(RENAME=(cumvar=&inparam.0_t2_v));
  SET &indata.;
  BY studyid ppcat treatment usubjid start;
  RETAIN cumvar;
  IF FIRST.usubjid THEN cumvar=0;
  IF &inparam.t1_t2_v NE . AND cumvar NE . THEN cumvar=(cumvar+
&inparam.t1_t2_v);
  ELSE IF &inparam.t1_t2_v EQ . OR cumvar EQ . THEN cumvar=.;

  &inparam.0_t2_u=&inparam.t1_t2_u;
RUN;
%MEND cumulativ;

%cumulativ(indata=values, inparam=ae);
%cumulativ(indata=cumae,  inparam=fe);
```

For scientific reasons, the cumulative renal clearance values cannot be calculated using the *cumulativ* macro. Since the values cannot simply be summed, the *cumclr* data file derives the values and assignment of units for renal CLR.

```
DATA cumclr;
  SET cumfe;
  clr0_t2_v=(ae0_t2_v/auc0_t2_v)/60;
  clr0_t2_u=clrt1_t2_u;
RUN;
```

Another user defined macro, *makefinal*, takes the final data, keeps the required variables and applies formats, creating a standardized data file format. The data file can then be used for further processing within globally harmonized software systems. Lastly, the urine endpoints are sorted and merged by the afore mentioned criteria and exclusion flags are assigned.

```
LET finalds=empty;
%MACRO makefinal(inparam=);
  DATA final&inparam.(KEEP=studyid usubjid period treatment
visitnum ppcat pptest ppscat ppstresn ppstresu dosec dosecu);
    FORMAT ppstresu pptest ppscat $20. ppstresn 32.20;
    SET prefinal(RENAME=(&inparam.=pptest &inparam._v=ppstresn
&inparam._u=ppstresu)
                 WHERE=(pptest NE ''));
    ppscat='CALC';
  RUN;

  %IF &finalds=empty %THEN %LET finalds=final&inparam.;
  %ELSE %LET finalds=%TRIM(%LEFT(&finalds.)) final&inparam.;
%MEND makefinal;


* Call the macro per type of parameter *;
%makefinal(inparam=aet1_t2);
%makefinal(inparam=ae0_t2);

%makefinal(inparam=fet1_t2);
%makefinal(inparam=fe0_t2);

%makefinal(inparam=clrt1_t2);
%makefinal(inparam=clr0_t2);

%makefinal(inparam=rae0_t2tr);
%makefinal(inparam=rae0_t2met);
%makefinal(inparam=rae0_t2mtmr);


DATA finalall;
  SET &finalds.;
 RUN;
```

CONCLUSION

The authors of this paper introduced and described a new approach to calculate PK urine endpoints using SAS® programming. The program template includes a combination of DATA steps and a user defined macros, thus allowing the programmer to easily adapt the code for the requirements of specific clinical trial endpoints. Several files can be created, consolidated and formatted in a final data file containing PK urine endpoints with exclusion criteria. The SAS® program successfully calculated the PK urine endpoints, see *Finalall* data file results below. Actual number of created data files is dependent on PK urine endpoints requested and sampling intervals provided in the rawdata file. While the initial time invested into the set-up of the program could be cumbersome, the programmer saves a significant amount of time if a version update of the PK analysis data set occurs. In addition, using SAS® programming to derive PK urine endpoints reduces errors commonly found when conducting manual calculations. This also reduces the amount of time for quality checking. One of the benefits of using the approach outlined in this paper is its ability to import PK urine raw data directly into SAS® to calculate endpoints without the use of WNL®. In conclusion, the program template saves time, resource allocation, expedites processes and increases data quality.

Finalall data file results

KPU	KPNM	PKALG	KP	PTNO	PKPER	ACTEVENT	PKTRT	PKMATRIX	DOSEC	DOSECU
ng	Ae0_12_ss	CALC	91842.000000000000000000	371002	1	113.00	A	URINE	50.000000	mg
ng	Ae0_24_ss	CALC	137630.400000000000000000	371002	1	113.00	A	URINE	50.000000	mg
ng	Ae0_4_ss	CALC	40732.000000000000000000	371002	1	113.00	A	URINE	50.000000	mg
ng	Ae12_24_ss	CALC	45788.400000000000000000	371002	1	113.00	A	URINE	50.000000	mg
ng	Ae4_12_ss	CALC	51110.000000000000000000	371002	1	113.00	A	URINE	50.000000	mg
mL/min	CLR0_24_ss	CALC	22.130659535983900000	371002	1	113.00	A	URINE	50.000000	mg
mL/min	CLR0_4_ss	CALC	18.890294659975800000	371002	1	113.00	A	URINE	50.000000	mg
mL/min	CLR12_24_ss	CALC	22.784012613740500000	371002	1	113.00	A	URINE	50.000000	mg
mL/min	CLR4_12_ss	CALC	24.894300590484000000	371002	1	113.00	A	URINE	50.000000	mg
mL/min	CLRss	CALC	21.818726577438900000	371002	1	113.00	A	URINE	50.000000	mg
%	fe0_12_ss	CALC	0.183684000000000000	371002	1	113.00	A	URINE	50.000000	mg
%	fe0_24_ss	CALC	0.275260800000000000	371002	1	113.00	A	URINE	50.000000	mg
%	fe0_4_ss	CALC	0.081464000000000000	371002	1	113.00	A	URINE	50.000000	mg
%	fe12_24_ss	CALC	0.091576800000000000	371002	1	113.00	A	URINE	50.000000	mg
%	fe4_12_ss	CALC	0.102220000000000000	371002	1	113.00	A	URINE	50.000000	mg
ng	Ae0_12_ss	CALC	102833.000000000000000000	371003	1	113.00	A	URINE	50.000000	mg
ng	Ae0_24_ss	CALC	153353.600000000000000000	371003	1	113.00	A	URINE	50.000000	mg
ng	Ae0_4_ss	CALC	44697.000000000000000000	371003	1	113.00	A	URINE	50.000000	mg
ng	Ae12_24_ss	CALC	50520.600000000000000000	371003	1	113.00	A	URINE	50.000000	mg
ng	Ae4_12_ss	CALC	58136.000000000000000000	371003	1	113.00	A	URINE	50.000000	mg
mL/min	CLR0_24_ss	CALC	16.511840217741800000	371003	1	113.00	A	URINE	50.000000	mg
mL/min	CLR0_4_ss	CALC	20.382599116855300000	371003	1	113.00	A	URINE	50.000000	mg
mL/min	CLR12_24_ss	CALC	13.318032265501200000	371003	1	113.00	A	URINE	50.000000	mg
mL/min	CLR4_12_ss	CALC	17.610593467914700000	371003	1	113.00	A	URINE	50.000000	mg

ACKNOWLEDGEMENTS

The authors would like to thank Tom MacGregor for reviewing the paper and providing input in conjunction with Katja Boland, James Hilbert and Peter Stopfer for their support.

CONTACT INFORMATION

Vanessa Rubano
Boehringer Ingelheim Pharmaceuticals Inc.
Ridgefield, Connecticut USA
vanessa.rubano@boehringer-ingelheim.com

Modesta Wiersema
Boehringer Ingelheim Pharma GmbH & Co. KG
Biberach an der Riß, Germany
modesta.wiersema@boehringer-ingelheim.com